# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>
# Contributor: Gaetan Bisson <bisson@archlinux.org>
# Contributor: Alexander Fehr <pizzapunk gmail com>
# Contributor: Giovanni Scafora <giovanni@archlinux.org>

pkgname=audacious
pkgver=4.3.1
pkgrel=1
pkgdesc="Lightweight, advanced audio player focused on audio quality"
arch=('loong64' 'x86_64')
url="https://audacious-media-player.org/"
license=('BSD')
depends=('qt5-base' 'qt5-svg' 'qt5-imageformats' 'glib2' 'hicolor-icon-theme'
         'desktop-file-utils')
makedepends=('meson')
optdepends=('unzip: zipped skins support')
source=(https://distfiles.audacious-media-player.org/$pkgname-$pkgver.tar.bz2)
sha256sums=('85e9e26841505b51e342ee72a2d05f19bef894f567a029ebb3f3e0c1adb42042')

build() {
  arch-meson $pkgname-$pkgver build \
    -Dgtk=false \
    -Dbuildstamp='Arch Linux'
  meson compile -C build
}

package() {
  depends+=('audacious-plugins')

  meson install -C build --destdir "$pkgdir"
  install -Dm644 $pkgname-$pkgver/contrib/audacious.appdata.xml -t "$pkgdir/usr/share/metainfo"
  install -Dm644 $pkgname-$pkgver/COPYING -t "$pkgdir/usr/share/licenses/$pkgname"
}

# vim:set ts=2 sw=2 et:
